Catenaries

Consider a perfectly flexible string or a chain with uniformly distributed load which we suspend by their ends.

a) Determine the function which describes the equilibrium figure of the string if the string has to bear only its own weight.

b) Determine the arc length above the [0,x] interval of the string as the function of the x.

This issue perfectly illustrates the relationship between the continuous and discrete modelling. On the one hand, the string can be considered as a solid with continuously distributed load. On the other hand, it can be considered as a curve consisting of particles in which case the connecting parts can only affect each other through the adjacent elements. A small part of the string transfers the pulling force affecting along the tangent of the curve to the adjacent part which the string resists thus does not stretch. However, the parts are unable to resist the cross directional force, the twisting force and the thrust. The figure of the string changes when it is affected by these forces.

The equilibrium figure of the string can be seen below. This curve is called the catenary.



In the figure the deepest point of the sag was denoted with the A and it was entered on the y axis from a 0<a distance from the x axis. The coordinates of another, arbitrary P point of the string was denoted with [x,y] and the length of the AP string part was denoted with s. If the w is the weight of a unit long part of the string then the weight of the AP arc of the string is ws. Three different forces are applied on the AP arc:

- the ws weight force is applied vertically downwards on the centre of the AP arc of the string
- the H pulling force in the A point, which is transferred by the string parts because of the left side suspension, points towards the negative direction of the x axis
- the P point is the striking point of the T holding force and its direction is the tangent of the string which forms a phi angle with the x axis.

Let’s start the solution of the task by writing the assumption of the balance of force for the x and y directional components separately.

> *restart*

> $T\cos(\text{phi}) = H$

$$T\cos(\phi) = H \quad (1)$$

> $T\sin(\text{phi}) = w\,s$

$$T\sin(\phi) = w\,s \quad (2)$$

After we express the T from the (1) and substitute it to the (2) we can easily determine the slope of the curve we are looking for.

> $isolate(T\cos(\text{phi}) = H, T)$

$$T = \frac{H}{\cos(\phi)} \quad (3)$$

> $subs(\%, T\sin(\text{phi}) = w\,s)$

$$\frac{H\sin(\phi)}{\cos(\phi)} = w\,s \quad (4)$$

> $\frac{\%}{H}$

$$\frac{\sin(\phi)}{\cos(\phi)} = \frac{w\,s}{H} \quad (5)$$

> $meredekség := student_{powsubs}\left(\frac{w}{H} = \frac{1}{a}, (5)\right)$

$$meredekség := \frac{\sin(\phi)}{\cos(\phi)} = \frac{s}{a} \quad (6)$$

For the sake of clarity, the a=H/w notation has been introduced.

So we have found a relation between the slope of the function we are looking for and the s arc length. But it is known that the slope, the substitution value of the differential and the arc length can be expressed by the well-known formula

$\int_0^x \sqrt{\left(1 + \frac{\partial}{\partial x} y\right)^2}\, dt$ If we do these substitutions in the equalities received for the slope then we get the differential equation concerning the y=y(x) function.

> $ivhossz := s = \int_0^x \sqrt{1 + \left(\frac{d}{dt} y(t)\right)^2}\, dt$

$$ivhossz := s = \int_0^x \sqrt{1 + \left(\frac{d}{dt} y(t)\right)^2}\, dt \quad (7)$$

> $subs(ivhossz, meredekség)$

(8)

$$\frac{\sin(\phi)}{\cos(\phi)} = \frac{\int_0^x \sqrt{1+\left(\frac{d}{dt}\,y(t)\right)^2}\,dt}{a} \qquad \textbf{(8)}$$

> $subs\left(\frac{\sin(phi)}{\cos(phi)} = \frac{d}{dx}\,y(x), \%\right)$

$$\frac{d}{dx}\,y(x) = \frac{\int_0^x \sqrt{1+\left(\frac{d}{dt}\,y(t)\right)^2}\,dt}{a} \qquad \textbf{(9)}$$

We have received a first-order differential equation on the right side of which there is an integral. To get rid of this, let’s differentiate both sides of the first-order differential equation by the x thus we get a second-order, nonlinear differential equation which is partial in the x and can be solved by the dsolve procedure.

> $\frac{d}{dx}\left(\textbf{(9)}\right)$

$$\frac{d^2}{dx^2}\,y(x) = \frac{\sqrt{1+\left(\frac{d}{dx}\,y(x)\right)^2}}{a} \qquad \textbf{(10)}$$

> $dsolve(\%, y(x))$

$$y(x) = -\mathrm{I}\,x + \_C1,\ y(x) = \mathrm{I}\,x + \_C1,\ y(x) = a\cosh\left(\frac{x+\_C1}{a}\right) + \_C2 \qquad \textbf{(11)}$$

> $megoldás := \%_3$

$$megoldás := y(x) = a\cosh\left(\frac{x+\_C1}{a}\right) + \_C2 \qquad \textbf{(12)}$$

To solve the task we have to determine the value of the _C1 and _C2 free parameters appearing in the general solution of the differential equation.

The value of the derivative of the function is 0 at the x=0 because the tangent of the curve is parallel with the x axis in the A point.

> $\frac{d}{dx}\,megoldás$

$$\frac{d}{dx}\,y(x) = \sinh\left(\frac{x+\_C1}{a}\right) \qquad \textbf{(13)}$$

> $subs\left(\frac{d}{dx}\,y(x) = 0, x = 0, \%\right)$

$$0 = \sinh\left(\frac{\_C1}{a}\right) \qquad \textbf{(14)}$$

> $C_1 := \_C1 = solve(\%, \_C1)$

(15)

$$C_1 := \_C1 = 0 \tag{15}$$

Notice that the diff procedure disregarded the fact that it had received an inequality as a first operand. It did the derivation on both sides of the equality. After this we substituted zero into the place of the derivative of the x and the function and solved the equality for _C1.

Let's substitute the _C1=0 value to the "solution" and simplify the expression.

> $megoldás := subs(C_1, megoldás)$

$$megoldás := y(x) = a \cosh\left(\frac{x}{a}\right) + \_C2 \tag{16}$$

Without any constraints on the generalities we can assume that the y(0)=a will be true. This can be achieved by shifting the coordinate system. In this way we get the zero value for the _C2 which we can substitute into the solution and the get the function describing the catenary.

> $subs(x = 0, y(0) = a, megoldás)$

$$a = a \cosh(0) + \_C2 \tag{17}$$

> $C_2 := \_C2 = solve(\%, \_C2)$

$$C_2 := \_C2 = 0 \tag{18}$$

> $láncgörbe := subs(C_2, megoldás)$

$$láncgörbe := y(x) = a \cosh\left(\frac{x}{a}\right) \tag{19}$$

Let's draw the catenary by the a=1 parameter value.

> $plot(subs(a = 1, rhs(láncgörbe)), x = -2..2, y = 0..3, scaling = constrained)$



Here is the exponential syntax of the catenary for the sake of those who are interested. This syntax is provided by the exp option of the convert procedure.

> *lhs*(*láncgörbe*) = *convert*(*rhs*(*láncgörbe*), exp)

$$y(x) = a\left(\frac{1}{2}\,\mathrm{e}^{\left(\frac{x}{a}\right)} + \frac{1}{2}\,\frac{1}{\mathrm{e}^{\left(\frac{x}{a}\right)}}\right) \qquad \textbf{(20)}$$

The solution of the b part of the task, that is, the determination of the arc length of the catenary is still ahead. Naturally it is as easy as pie. We have the formula with which we can calculate the arc length so we only have to substitute the indefinite y(t) located in it with the function that describes the catenary.

> *ivhossz*

$$s = \int_0^x \sqrt{1 + \left(\frac{\mathrm{d}}{\mathrm{d}t}\,y(t)\right)^2}\,\mathrm{d}t \qquad \textbf{(21)}$$

> *lánc* := *unapply*(*rhs*(*láncgörbe*), *x*)

$$lánc := x \rightarrow a\cosh\left(\frac{x}{a}\right) \qquad \textbf{(22)}$$

> $subs\left(y(t) = lánc(t), s = \int_0^x \sqrt{1 + \left(\frac{d}{dt} y(t)\right)^2}\, dt\right)$

$$s = \int_0^x \sqrt{1 + \left(\frac{\partial}{\partial t}\left(a \cosh\left(\frac{t}{a}\right)\right)\right)^2}\, dt \quad \textbf{(23)}$$

> $assuming([eval(\%)], [positive])$

$$s = \frac{1}{2} e^{\left(-\frac{x}{a}\right)} a \left(e^{\left(\frac{2x}{a}\right)} - 1\right) \quad \textbf{(24)}$$

We would like to highlight the importance of the second instruction because the “catenary” is an expression so we had to create a function from it before the substitution. What would be the result if we skipped this step?

Finally, the trig option of the convert procedure gives the hyperbolic syntax of the arc length which we simplify within two steps.

> $convert(\textbf{(24)}, trig)$

$$s = \frac{1}{2}\left(\cosh\left(\frac{x}{a}\right) - \sinh\left(\frac{x}{a}\right)\right) a \left(\cosh\left(\frac{2x}{a}\right) + \sinh\left(\frac{2x}{a}\right) - 1\right) \quad \textbf{(25)}$$

> $expand(\%)$

$$s = a \cosh\left(\frac{x}{a}\right)^3 - a \cosh\left(\frac{x}{a}\right) + a \sinh\left(\frac{x}{a}\right) - a \sinh\left(\frac{x}{a}\right)^2 \cosh\left(\frac{x}{a}\right) \quad \textbf{(26)}$$

> $simplify(\%)$

$$s = a \sinh\left(\frac{x}{a}\right) \quad \textbf{(27)}$$

We finish this worksheet with the graph of the function that describes the arc length.

> $assign(\%)$

```
> plot(subs(a=1,s), x=0..2, y=0..3, scaling=constrained, title=`A
  láncgörbe ívhosszának változása`);
```



>

What Have You Learnt About Maple?

- The convert(expression, exp) and the convert(expression, trig) instructions can be used to create the exponential syntax of the trigonometric and hyperbolic functions and vice versa

Exercises

1. Prove that $y(x)^2 - s(x)^2 = a^2$, in which case the y(x) is the value of the catenary at the place of the x while the s(x) denotes the arc length at the [0,x] interval.

2. Make an animated illustrative procedure of the changing of the a parameter of the catenary. How does the figure of the catenary change by changing the a?

3. We saw that a=H/w. Based on this prove that if we increase the weight of the unit string length then the sag also increases.

4. Assume that the length of a string is 2 l and its total weight is w. Assume that both ends of the string are suspended from the P and Q points at the same height. The lowest point of the string should be at h distance under the PQ line. Prove that the [képlet] equality is true for the a parameter of the catenary. Deduce that the extent of the pulling force that affects at the P or Q point is

$$\frac{1}{4} w \cdot \left(\frac{l}{h} + \frac{h}{l} \right).$$